

NAG Fortran Library Routine Document

G13DCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of **bold italicised** terms and other implementation-dependent details.

1 Purpose

G13DCF fits a vector autoregressive moving average (VARMA) model to an observed vector of time series using the method of maximum likelihood. Standard errors of parameter estimates are computed along with their appropriate correlation matrix. The routine also calculates estimates of the residual series.

2 Specification

```

SUBROUTINE G13DCF(K, N, IP, IQ, MEAN, PAR, NPAR, QQ, IK, W, PARHLD,
1          EXACT, IPRINT, CGETOL, MAXCAL, ISHOW, NITER, RLOGL, V,
2          G, CM, ICM, WORK, LWORK, IW, LIW, IFAIL)
    INTEGER      K, N, IP, IQ, NPAR, IK, IPRINT, MAXCAL, ISHOW, NITER,
1          ICM, LWORK, IW(LIW), LIW, IFAIL
    real        PAR(NPAR), QQ(IK,K), W(IK,N), CGETOL, RLOGL, V(IK,N),
1          G(NPAR), CM(ICM,NPAR), WORK(LWORK)
    LOGICAL      MEAN, PARHLD(NPAR), EXACT

```

3 Description

Let $W_t = (w_{1t}, w_{2t}, \dots, w_{kt})^T$, for $t = 1, 2, \dots, n$, denote a vector of k time series which is assumed to follow a multivariate ARMA model of the form

$$W_t - \mu = \phi_1(W_{t-1} - \mu) + \phi_2(W_{t-2} - \mu) + \dots + \phi_p(W_{t-p} - \mu) + \epsilon_t - \theta_1\epsilon_{t-1} - \theta_2\epsilon_{t-2} - \dots - \theta_q\epsilon_{t-q} \quad (1)$$

where $\epsilon_t = (\epsilon_{1t}, \epsilon_{2t}, \dots, \epsilon_{kt})^T$, for $t = 1, 2, \dots, n$ is a vector of k residual series assumed to be Normally distributed with zero mean and positive-definite covariance matrix Σ . The components of ϵ_t are assumed to be uncorrelated at non-simultaneous lags. The ϕ_i and θ_j are k by k matrices of parameters. $\{\phi_i\}$, for $i = 1, 2, \dots, p$, are called the autoregressive (AR) parameter matrices, and $\{\theta_i\}$, for $i = 1, 2, \dots, q$, the moving average (MA) parameter matrices. The parameters in the model are thus the p (k by k) ϕ -matrices, the q (k by k) θ -matrices, the mean vector, μ , and the residual error covariance matrix Σ . Let

$$A(\phi) = \begin{bmatrix} \phi_1 & I & 0 & \dots & \dots & 0 \\ \phi_2 & 0 & I & 0 & \dots & 0 \\ \vdots & & & \ddots & & \\ \vdots & & & & \ddots & \\ \phi_{p-1} & 0 & \dots & \dots & \dots & 0 & I \\ \phi_p & 0 & \dots & \dots & \dots & 0 & 0 \end{bmatrix}_{pk \times pk} \quad \text{and} \quad B(\theta) = \begin{bmatrix} \theta_1 & I & 0 & \dots & \dots & 0 \\ \theta_2 & 0 & I & 0 & \dots & 0 \\ \vdots & & & \ddots & & \\ \vdots & & & & \ddots & \\ \vdots & & & & & \ddots & \\ \theta_{q-1} & 0 & \dots & \dots & \dots & \dots & I \\ \theta_q & 0 & \dots & \dots & \dots & \dots & 0 \end{bmatrix}_{qk \times qk}$$

where I denotes the k by k identity matrix.

The model (1) is said to be stationary if the eigenvalues of $A(\phi)$ lie inside the unit circle. Similarly, (1) is said to be invertible if the eigenvalues of $B(\theta)$ lie inside the unit circle.

The method of computing the exact likelihood function (using a Kalman filter algorithm) is discussed in Shea (1987). A quasi-Newton algorithm (see Gill and Murray (1972)) is then used to search for the maximum of the log-likelihood function. Stationarity and invertibility are enforced on the model using the reparameterisation discussed in Ansley and Kohn (1986). Conditional on the maximum likelihood estimates being equal to their true values the estimates of the residual series are uncorrelated with zero mean and constant variance Σ .

The user has the option of setting a parameter (EXACT to .FALSE.) so that G13DCF calculates conditional maximum likelihood estimates (conditional on $W_0 = W_{-1} = \dots = W_{1-p} = \epsilon_0 = \epsilon_{-1} = \dots = \epsilon_{1-q} = 0$). This may be useful if the exact maximum likelihood estimates are close to the boundary of the invertibility region.

The user also has the option (see Section 5) of requesting G13DCF to constrain elements of the ϕ and θ matrices and μ vector to have pre-specified values.

4 References

Ansley C F and Kohn R (1986) A note on reparameterising a vector autoregressive moving average model to enforce stationarity *J. Statist. Comput. Simulation* **24** 99–106

Gill P E and Murray W (1972) Quasi-Newton methods for unconstrained optimization *J. Inst. Math. Appl.* **9** 91–108

Shea B L (1987) Estimation of multivariate time series *J. Time Ser. Anal.* **8** 95–110

5 Parameters

- 1: K – INTEGER *Input*
On entry: the number of observed time series, k .
Constraint: $K \geq 1$.
- 2: N – INTEGER *Input*
On entry: the number of observations in each time series, n .
Constraint: $N \geq 3$.
- 3: IP – INTEGER *Input*
On entry: the number of AR parameter matrices, p .
Constraint: $IP \geq 0$.
- 4: IQ – INTEGER *Input*
On entry: the number of MA parameter matrices, q .
Constraint: $IQ \geq 0$.
Note: $IP = IQ = 0$ is **not permitted**.
- 5: MEAN – LOGICAL *Input*
On entry: MEAN must be set to .TRUE. if components of μ are to be estimated and .FALSE. if all elements of μ are to be taken as zero.
- 6: PAR(NPAR) – *real* array *Input/Output*
On entry: initial parameter estimates read in row by row in the order $\phi_1, \phi_2, \dots, \phi_p, \theta_1, \theta_2, \dots, \theta_q, \mu$. Thus, if $IP > 0$ then $PAR((l-1) \times k \times k + (i-1) \times k + j)$ must be set equal to an initial estimate of the (i, j) th element of ϕ_l for $l = 1, 2, \dots, p$; $i, j = 1, 2, \dots, k$. If $IQ > 0$, then $PAR(p \times k \times k + (l-1) \times k \times k + (i-1) \times k + j)$ must be set equal to an initial estimate of the (i, j) th element of θ_l , $l = 1, 2, \dots, q$; $i, j = 1, 2, \dots, k$. If MEAN has been set equal to .TRUE., then $PAR((p+q) \times k \times k + i)$ should be set equal to an initial estimate of the i th component of μ ($\mu(i)$). (If the user sets $PAR((p+q) \times k \times k + i)$ to 0.0 then G13DCF will calculate the mean of the i th series and use this as an initial estimate of $\mu(i)$.)
The first $p \times k \times k$ elements of PAR must satisfy the stationarity condition and the next $q \times k \times k$ elements of PAR must satisfy the invertibility condition.
If in doubt set all elements of PAR to 0.0.

On exit: if $IFAIL = 0$ or $IFAIL \geq 4$ then all the elements of PAR will be overwritten by the latest estimates of the corresponding ARMA parameters.

- 7: NPAR – INTEGER *Input*
On entry: the number of initial parameter estimates. The total number of observations ($n \times k$) must exceed the total number of parameters in the model $(NPAR + k(k + 1)/2)$.
Constraint: if $MEAN = .FALSE.$, NPAR must be set equal to $(p + q) \times k \times k$; if $MEAN = .TRUE.$, $(p + q) \times k \times k + k$.
- 8: QQ(IK,K) – *real* array *Input/Output*
On entry: $QQ(i, j)$ must be set equal to an initial estimate of the (i, j) th element of Σ . The lower triangle only is needed. QQ must be positive-definite. It is strongly recommended that on entry the elements of QQ are of the same order of magnitude as at the solution point. If the user sets $QQ(i, j) = 0.0$, for $i = 1, 2, \dots, k$; $j = 1, 2, \dots, i$, then G13DCF will calculate the covariance matrix between the k time series and use this as an initial estimate of Σ .
On exit: if $IFAIL = 0$ or $IFAIL \geq 4$ then $QQ(i, j)$ will contain the latest estimate of the (i, j) th element of Σ . The lower triangle only is returned.
- 9: IK – INTEGER *Input*
On entry: the first dimension of the arrays QQ, W and V as declared in the (sub)program from which G13DCF is called.
Constraint: $IK \geq K$.
- 10: W(IK,N) – *real* array *Input*
On entry: $W(i, t)$ must be set equal to the i th component of W_t , for $i = 1, 2, \dots, k$; $t = 1, 2, \dots, n$.
- 11: PARHLD(NPAR) – LOGICAL array *Input*
On entry: $PARHLD(i)$ must be set to $.TRUE.$, if $PAR(i)$ is to be held constant at its input value and $.FALSE.$, if $PAR(i)$ is a free parameter, for $i = 1, 2, \dots, NPAR$.
 If in doubt try setting all elements of PARHLD to $.FALSE.$.
- 12: EXACT – LOGICAL *Input*
On entry: EXACT must be set equal to $.TRUE.$ if the user wishes the routine to compute exact maximum likelihood estimates. EXACT must be set equal to $.FALSE.$ if only conditional likelihood estimates are required.
- 13: IPRINT – INTEGER *Input*
On entry: the frequency with which the automatic monitoring routine is to be called. See Section 9 for an example of the printed output.
 If $IPRINT > 0$, the ML search procedure is monitored once every IPRINT iterations and just before exit from the search routine.
 If $IPRINT=0$, the search routine is monitored once at the final point.
 If $IPRINT < 0$ the search routine is not monitored at all.
- 14: CGETOL – *real* *Input*
On entry: the accuracy to which the solution in PAR and QQ is required. If CGETOL is set to 10^{-l} and on exit $IFAIL=0$ or $IFAIL \geq 6$, then all the elements in PAR and QQ should be accurate to approximately l decimal places. For most practical purposes the value 10^{-4} should suffice. The user should be wary of setting CGETOL too small since the convergence criteria may then have become too strict for the machine to handle. If CGETOL has been set to a value which is less than the *machine precision*, ϵ , then G13DCF will use the value $10.0 \times \sqrt{\epsilon}$ instead.

- 15: MAXCAL – INTEGER *Input*
On entry: the maximum number of likelihood evaluations to be permitted by the search procedure.
Constraint: MAXCAL \geq 1.
Suggested value: MAXCAL = $40 \times \text{NPAR} \times (\text{NPAR} + 5)$.
- 16: ISHOW – INTEGER *Input*
On entry: specifies which of the following two quantities are to be printed.
 (a) Table of maximum likelihood estimates and their standard errors (as returned in the output arrays PAR, QQ and CM).
 (b) Table of residual series (as returned in the output array V).
 ISHOW=0 none of the above are printed,
 ISHOW=1 (a) only is printed,
 ISHOW=2 (a) and (b) are printed.
Constraint: $0 \leq \text{ISHOW} \leq 2$.
- 17: NITER – INTEGER *Output*
On exit: if IFAIL = 0 or IFAIL \geq 4 then NITER contains the number of iterations performed by the search routine.
- 18: RLOGL – *real* *Output*
On exit: if IFAIL = 0 or IFAIL \geq 4 then RLOGL contains the value of the log-likelihood function corresponding to the final point held in PAR and QQ.
- 19: V(IK,N) – *real* array *Output*
On exit: if IFAIL = 0 or IFAIL \geq 4 then V(*i*, *t*) will contain an estimate of the *i*th component of ϵ_t , for $i = 1, 2, \dots, k$; $t = 1, 2, \dots, n$, corresponding to the final point held in PAR and QQ.
- 20: G(NPAR) – *real* array *Output*
On exit: if IFAIL = 0 or IFAIL \geq 4 then G(*i*) will contain the estimated first derivative of the log-likelihood function with respect to the *i*th element in the array PAR. If the gradient cannot be computed then all the elements of G are returned as zero.
- 21: CM(ICM,NPAR) – *real* array *Output*
On exit: if IFAIL = 0 or IFAIL \geq 4 then CM(*i*, *j*) will contain an estimate of the correlation coefficient between the *i*th and *j*th elements in the PAR array for $1 \leq i \leq \text{NPAR}$, $1 \leq j \leq \text{NPAR}$. If $i = j$ then CM(*i*, *j*) will contain the estimated standard error of PAR(*i*). If the *l*th component of PAR has been held constant, i.e., PARHLD(*l*) was set to .TRUE., then the *l*th row and column of CM will be set to zero. If the second derivative matrix cannot be computed then all the elements of CM are returned as zero.
- 22: ICM – INTEGER *Input*
On entry: the first dimension of the array CM as declared in the (sub)program from which G13DCF is called.
Constraint: ICM \geq NPAR.

- 23: WORK(LWORK) – *real* array *Workspace*
 24: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which G13DCF is called.

Constraints:

$$\text{if } r = \max(p, q) \text{ and } M = \text{NPAR} + k(k+1)/2, \text{ then}$$

$$\text{LWORK} \geq k(5 + 3n + 2r) + k^2(2p + q + 2r(r+2) + 9) + M(5M + 29)/2 + k^4(r+1)^2.$$

- 25: IW(LIW) – INTEGER array *Workspace*
 26: LIW – INTEGER *Input*

On entry: the dimension of the array IW as declared in the (sub)program from which G13DCF is called.

Constraint: $\text{LIW} \geq \text{NPAR} + kr + 2 + k(k+1)/2$.

- 27: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL \neq 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

- On entry, $K < 1$,
- or $IP < 0$,
- or $IQ < 0$,
- or $IP=IQ=0$,
- or $\text{NPAR} \neq (IP + IQ) \times K \times K + \Delta \times K$, where $\Delta = 1$ if MEAN = .TRUE. or $\Delta = 0$ if MEAN = .FALSE.,
- or $N \times K \leq \text{NPAR} + K \times (K + 1)/2$,
- or $IK < K$,
- or $\text{MAXCAL} < 1$,
- or $\text{ISHOW} < 0$,
- or $\text{ISHOW} > 2$,
- or $\text{ICM} < \text{NPAR}$,
- or LWORK is too small,
- or LIW is too small.

IFAIL = 2

On entry, either the initial estimate of Σ is not positive-definite, or the initial estimates of the AR parameters are such that the model is non-stationary, or the initial estimates of the MA parameters are such that the model is non-invertible. To proceed, the user must try a different starting point.

IFAIL = 3

The routine cannot compute a sufficiently accurate estimate of the gradient vector at the user-supplied starting point. This usually occurs if either the initial parameter estimates are very close to the ML parameter estimates, or the user has supplied a very poor estimate of Σ or the starting point is very close to the boundary of the stationarity or invertibility region. To proceed, the user must try a different starting point.

IFAIL = 4

There have been MAXCAL log-likelihood evaluations made in the routine. If steady increases in the log-likelihood function were monitored up to the point where this exit occurred, then the exit probably simply occurred because MAXCAL was set too small, so the calculations should be restarted from the final point held in PAR and QQ. This type of exit may also indicate that there is no maximum to the likelihood surface. Output quantities (as described in Section 5) are computed at the final point held in PAR and QQ, except that if G or CM cannot be computed, they are set to zero.

IFAIL = 5

The conditions for a solution have not all been met, but a point at which the log-likelihood took a larger value could not be found.

Provided that the estimated first derivatives are sufficiently small, and that the estimated condition number of the second derivative (Hessian) matrix, as printed when IPRINT ≥ 0 , is not too large, this error exit may simply mean that, although it has not been possible to satisfy the specified requirements, the algorithm has in fact found the solution as far as the accuracy of the machine permits.

Such a condition can arise, for instance, if CGETOL has been set so small that rounding error in evaluating the likelihood function makes attainment of the convergence conditions impossible.

If the estimated condition number at the final point is large, it could be that the final point is a solution but that the smallest eigenvalue of the Hessian matrix is so close to zero at the solution that it is not possible to recognise it as a solution. Output quantities (as described in Section 5) are computed at the final point held in PAR and QQ, except that if G or CM cannot be computed, they are set to zero.

IFAIL = 6

The ML solution is so close to the boundary of either the stationarity region or the invertibility region that G13DCF cannot evaluate the Hessian matrix. The elements of CM will then be set to zero on exit. The elements of G will also be set to zero. All other output quantities will be correct.

IFAIL = 7

This is an unlikely exit, which could occur in E04XAF/E04XAA, which computes an estimate of the second derivative matrix and the gradient vector at the solution point. Either the Hessian matrix was found to be too ill-conditioned to be evaluated accurately or the gradient vector could not be computed to an acceptable degree of accuracy. In this case the elements of CM will be set to zero on exit as will the elements of G. All other output quantities will be correct.

IFAIL = 8

The second derivative matrix at the solution point is not positive-definite. In this case the elements of CM will be set to zero on exit. All other output quantities will be correct.

7 Accuracy

On exit from G13DCF, if IFAIL = 0 or IFAIL ≥ 6 and CGETOL has been set to 10^{-l} , then all the parameters should be accurate to approximately l decimal places. If CGETOL was set equal to a value less than the *machine precision*, ϵ , then all the parameters should be accurate to approximately $10.0 \times \sqrt{\epsilon}$.

If IFAIL = 4 on exit (i.e., MAXCAL likelihood evaluations have been made but the convergence conditions of the search routine have not been satisfied), then the elements in PAR and QQ may still be good approximations to the ML estimates. The user is advised to inspect the elements of G to see whether this is likely to be so.

8 Further Comments

8.1 Timing

The number of iterations required depends upon the number of parameters in the model and the distance of the user-supplied starting point from the solution.

8.2 Constraining for Stationarity and Invertibility

If the solution lies on the boundary of the admissibility region (stationarity and invertibility region) then G13DCF may get into difficulty and exit with IFAIL=5. If this exit occurs the user is advised to either try a different starting point or a different setting for EXACT. If this still continues to occur then the user is urged to try fitting a more parsimonious model.

8.3 Over-parameterisation

The user is advised to try and avoid fitting models with an excessive number of parameters since over-parameterisation can cause the maximization problem to become ill-conditioned.

8.4 Standardising the Residual Series

The standardised estimates of the residual series ϵ_t (denoted by $\hat{\epsilon}_t$) can easily be calculated by forming the Cholesky decomposition of Σ , e.g., GG^T and setting $\hat{\epsilon}_t = G^{-1}\epsilon_t$. F07FDF (SPOTRF/DPOTRF) may be used to calculate the array G. The components of $\hat{\epsilon}_t$ which are now uncorrelated at **all** lags can sometimes be more easily interpreted.

8.5 Assessing the Fit of the Model

If the user's time series model provides a good fit to the data then the residual series should be approximately white noise, i.e., exhibit no serial cross-correlation. An examination of the residual cross-correlation matrices should confirm whether this is likely to be so. The user is advised to call G13DSF to provide information for diagnostic checking. G13DSF returns the residual cross-correlation matrices along with their asymptotic standard errors. The routine also computes a portmanteau statistic and its asymptotic significance level for testing model adequacy. If IFAIL=0 or IFAIL \geq 5 on exit from G13DCF then the output quantities K, N, V, IK, IP, IQ, PAR, PARHLD, and QQ will be suitable for input to G13DSF.

9 Example

A program to fit a bivariate AR(1) model to two series each of length 48. μ will be estimated and $\phi_1(2, 1)$ will be constrained to be zero.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G13DCF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          KMAX, IK, IPMAX, IQMAX, NMAX, NPARMX, ICM, LWORK,
+                    LIW
      PARAMETER        ( KMAX=3, IK=KMAX, IPMAX=3, IQMAX=3, NMAX=100,
+                    NPARMX=( IPMAX+IQMAX ) * KMAX * KMAX + KMAX, ICM=NPARMX,
+                    LWORK=2000, LIW=100 )
      INTEGER          NIN, NOUT
      PARAMETER        ( NIN=5, NOUT=6 )
```

```

*   .. Local Scalars ..
  real          CGETOL, RLOGL
  INTEGER       I, IFAIL, IP, IPRINT, IQ, ISHOW, J, K, MAXCAL, N,
+             NITER, NPAR
  LOGICAL      EXACT, MEAN
*   .. Local Arrays ..
  real          CM(ICM,NPARMX), G(NPARMX), PAR(NPARMX),
+             QQ(IK,KMAX), V(IK,NMAX), W(IK,NMAX), WORK(LWORK)
  INTEGER       IW(LIW)
  LOGICAL      PARHLD(NPARMX)
*   .. External Subroutines ..
  EXTERNAL     G13DCF, X04ABF
*   .. Executable Statements ..
  WRITE (NOUT,*) 'G13DCF Example Program Results'
*   Skip heading in data file
  READ (NIN,*)
  READ (NIN,*) K, IP, IQ, N, MEAN
  CALL X04ABF(1,NOUT)
  WRITE (NOUT,*)
  IF (K.GT.0 .AND. K.LE.KMAX .AND. IP.GE.0 .AND. IP.LE.IPMAX .AND.
+   IQ.GE.0 .AND. IQ.LE.IQMAX) THEN
    NPAR = (IP+IQ)*K*K
    IF (MEAN) NPAR = NPAR + K
    IF ((N.LE.NMAX) .AND. (NPAR.LE.NPARMX)) THEN
      DO 20 I = 1, NPAR
        PAR(I) = 0.0e0
        PARHLD(I) = .FALSE.
20      CONTINUE
*
*       Set all elements of Q to zero to use covariance matrix
*       between the K time series as the initial estimate of the
*       covariance matrix
*
      DO 60 J = 1, K
        DO 40 I = J, K
          QQ(I,J) = 0.0e0
40        CONTINUE
60      CONTINUE
      DO 80 I = 1, K
        READ (NIN,*) (W(I,J),J=1,N)
80      CONTINUE
      PARHLD(3) = .TRUE.
      EXACT = .TRUE.
*       ** Set IPRINT .GT. 0 to obtain intermediate output
      IPRINT = -1
      CGETOL = 0.0001e0
      MAXCAL = 40*NPAR*(NPAR+5)
      ISHOW = 2
      IFAIL = -1
*
      CALL G13DCF(K,N,IP,IQ,MEAN,PAR,NPAR,QQ,IK,W,PARHLD,EXACT,
+             IPRINT,CGETOL,MAXCAL,ISHOW,NITER,RLOGL,V,G,CM,
+             ICM,WORK,LWORK,IW,LIW,IFAIL)
*
      END IF
    END IF
  STOP
  END

```

9.2 Program Data

G13DCF Example Program Data

2 1 0 48 T

-1.490	-1.620	5.200	6.230	6.210	5.860	4.090	3.180
2.620	1.490	1.170	0.850	-0.350	0.240	2.440	2.580
2.040	0.400	2.260	3.340	5.090	5.000	4.780	4.110
3.450	1.650	1.290	4.090	6.320	7.500	3.890	1.580
5.210	5.250	4.930	7.380	5.870	5.810	9.680	9.070
7.290	7.840	7.550	7.320	7.970	7.760	7.000	8.350
7.340	6.350	6.960	8.540	6.620	4.970	4.550	4.810

4.750	4.760	10.880	10.010	11.620	10.360	6.400	6.240
7.930	4.040	3.730	5.600	5.350	6.810	8.270	7.680
6.650	6.080	10.250	9.140	17.750	13.300	9.630	6.800
4.080	5.060	4.940	6.650	7.940	10.760	11.890	5.850
9.010	7.500	10.020	10.380	8.150	8.370	10.730	12.140

9.3 Program Results

G13DCF Example Program Results

VALUE OF IFAIL PARAMETER ON EXIT FROM G13DCF = 0

VALUE OF LOG LIKELIHOOD FUNCTION ON EXIT = -0.20280E+03

MAXIMUM LIKELIHOOD ESTIMATES OF AR PARAMETER MATRICES

```

-----
PHI(1)    ROW-WISE :    0.802    0.065
                    (0.091) (0.102)

                    0.000    0.575
                    (0.000) (0.121)

```

MAXIMUM LIKELIHOOD ESTIMATE OF PROCESS MEAN

```

-----
                    4.271    7.825
                    (1.219) (0.776)

```

MAXIMUM LIKELIHOOD ESTIMATE OF SIGMA MATRIX

```

-----
                    2.964

                    0.637    5.380

```

RESIDUAL SERIES NUMBER 1

```

-----
T      1      2      3      4      5      6      7      8
V(T) -3.33  -1.24  5.75  1.27  0.32  0.11  -1.27  -0.73

T      9     10     11     12     13     14     15     16
V(T) -0.58  -1.26  -0.67  -1.13  -2.02  -0.57  1.24  -0.13

T     17     18     19     20     21     22     23     24
V(T) -0.77  -2.09  1.34  0.95  1.71  0.23  -0.01  -0.60

T     25     26     27     28     29     30     31     32
V(T) -0.68  -1.89  -0.77  2.05  2.11  0.94  -3.32  -2.50

T     33     34     35     36     37     38     39     40
V(T)  3.16  0.47  0.05  2.77  -0.82  0.25  3.99  0.20

T     41     42     43     44     45     46     47     48
V(T) -0.70  1.07  0.44  0.28  1.09  0.50  -0.10  1.70

```

RESIDUAL SERIES NUMBER 2

```

-----
T      1      2      3      4      5      6      7      8
V(T) -0.19  -1.20  -0.02  1.21  -1.62  -2.16  -1.63  -1.13

T      9     10     11     12     13     14     15     16
V(T) -1.34  -1.30  4.82  0.43  2.54  0.35  -2.88  -0.77

T     17     18     19     20     21     22     23     24
V(T)  1.02  -3.85  -1.92  0.13  -1.20  0.41  1.03  -0.40

```

T	25	26	27	28	29	30	31	32
V(T)	-1.09	-1.07	3.43	-0.08	9.17	-0.23	-1.34	-2.06
T	33	34	35	36	37	38	39	40
V(T)	-3.16	-0.61	-1.30	0.48	0.79	2.87	2.38	-4.31
T	41	42	43	44	45	46	47	48
V(T)	2.32	-1.01	2.38	1.29	-1.14	0.36	2.59	2.64
